

Structural time series models, output gap estimation

Exercise

Setup:

You are asked to replicate some of the results in Orphanides and van Norden 2002 (OvN hereinafter). We will not replicate real-time output gap estimates but only focus on what OvN call “final” estimates. Note that despite OvN stating in their 2002 paper that they use data from 1965 to 1997, they use data from 1947 to 2000! (You will never get results even close to theirs if you limit yourself to 65-97 data.)

Tasks:

1.)

Open the file `croushore_stark.wf1` to find the 2000Q2 vintage of real US GDP. Transform the time series into logs.

2.)

Detrend the log of real GDP using a linear trend. Do you obtain similar results as OvN in Table 1?

3.)

Detrend the log of real GDP using the HP filter. Do you obtain similar results as OvN in Table 1?

4.)

Detrend the log of real GDP using the Watson model. Do you obtain similar results as OvN in Table 1? How many times is the output gap significantly different from zero over the period 1966-1997? (Only use the Kalman smoother standard errors).

Hint 1: Instead of using a diffuse prior, we assume that the output gap was closed at the beginning of the sample, hence set the initial distribution of the state to:

$$\alpha_0 = [\text{lgdp00q2}_{1947Q1} \quad 0 \quad 0]' \text{ and } P_0 = \begin{bmatrix} 0.01 & 0 & 0 \\ 0 & 0.01 & 0 \\ 0 & 0 & 0.01 \end{bmatrix}.$$

Hint 2: In the MLE, use starting values $\delta^{(1)} = 0.008$, $\sigma_\eta^2 = \exp\{-10.5\}$, $\rho_1 = 1.42$, $\rho_2 = -0.45$, $\sigma_\epsilon^2 = \exp\{-9.7\}$.

5.)

Detrend the log of real GDP using the Harvey-Clark model. Do you obtain similar results as OvN in Table 1? How many times is the output gap significantly different from zero over the period 1966-1997? (Only use the Kalman smoother standard errors).

Hint 1: Instead of using a diffuse prior, we assume that the output gap was closed at the beginning of the sample, hence set the initial distribution of the state to:

$$\alpha_0 = [\text{lgdp00q2}_{1947Q1} \quad 0 \quad 0 \quad 0]' \text{ and } P_0 = \begin{bmatrix} 0.01 & 0 & 0 & 0 \\ 0 & 0.01 & 0 & 0 \\ 0 & 0 & 0.01 & 0 \\ 0 & 0 & 0 & 0.01 \end{bmatrix}.$$

Hint 2: In the MLE, use starting values $\sigma_\eta^2 = \exp\{-10.4\}$, $\sigma_v^2 = \exp\{-17\}$, $\rho_1 = 1.44$, $\rho_2 = -0.51$, $\sigma_\varepsilon^2 = \exp\{-9.8\}$.

Solution:

By using the programme given below we get graphs very similar to OvN and the mean, max and min of the output gap series are close to the values reported in OvN table 1.

Watson: 23 times (out of 128) significant output gap (using $\pm 2RMSE$): 1966Q1-1967Q1, 1968Q1-1968Q4, 1969Q1-1969Q2, 1973Q1-1973Q4, 1978Q3-1979Q1, 1982Q2-1983Q2.

Harvey-Clark: 13 times (out of 128) significant output gap (using $\pm 2RMSE$): 1966Q1-1966Q2, 1968Q2-1968Q3, 1969Q1, 1973Q1-1973Q2, 1978Q4, 1982Q2-1983Q2.

Programme:

```
'----- MAKE SURE YOU HAVE THE DATA FILE OPEN

smpl @all

'----- TAKING LOGS
genr lgdp00q2=log(gdp00q2)

'----- LINEAR DETRENDING
equation lintrend.ls lgdp00q2=c(1)+c(2)*@trend
genr lgdp00q2_lt=c(1) + c(2)*@trend
genr gap_lt=lgdp00q2-lgdp00q2_lt

'----- HP FILTERING
lgdp00q2.hpf(lambda=1600) lgdp00q2_hp
genr gap_hp=lgdp00q2-lgdp00q2_hp

'----- WATSON DETRENDING
'----- CREATING THE STATE SPACE MODEL
sspace watson
watson.append @signal lgdp00q2 = mu + z
watson.append @state mu = c(11) + mu(-1) + [var=exp(c(12))]
watson.append @state z = c(13)*z(-1) + c(14)*z1(-1) + [var=exp(c(15))]
watson.append @state z1 = z(-1)
'----- CREATING STARTING INITIAL STATE
vector(3) insta_w=0
insta_w(1) = lgdp00q2(1)
sym(3) invar_w=0
invar_w(1,1)=0.01
invar_w(2,2)=0.01
invar_w(3,3)=0.01
watson.append @mprior insta_w
watson.append @vprior invar_w
'----- SETTING OPTIMAL START VALUES
watson.append param c(11) 0.008 c(12) -10.5 c(13) 1.42 c(14) -0.45 c(15) -9.7
'----- ESTIMATE!
watson.ml(showopts)
'----- MAKE OUTPUT GAP
watson.makestates(t=smooth) *sw
watson.makestates(t=smoothse) *swse
genr gap_w = zsw
genr gap_wp2se=gap_w+2*zswse
genr gap_wm2se=gap_w-2*zswse
```

```

'----- HARVEY-CLARK DETRENDING
'----- CREATING THE STATE SPACE MODEL
sspace hacla
hacla.append @signal lgdp00q2 = mu + z
hacla.append @state mu = g(-1) + mu(-1) + [var=exp(c(21))]
hacla.append @state g = g(-1) + [var=exp(c(22))]
hacla.append @state z = c(23)*z(-1) + c(24)*z1(-1) + [var=exp(c(25))]
hacla.append @state z1 = z(-1)
'----- CREATING STARTING INITIAL STATE
vector(4) insta_hc=0
insta_hc(1) = lgdp00q2(1)
sym(4) invar_hc=0
invar_hc(1,1)=0.01
invar_hc(2,2)=0.01
invar_hc(3,3)=0.01
invar_hc(4,4)=0.01
hacla.append @mprior insta_hc
hacla.append @vprior invar_hc
'----- SETTING OPTIMAL START VALUES
hacla.append param c(21) -10.4 c(22) -17 c(23) 1.44 c(24) -0.51 c(25) -9.8
'----- ESTIMATE!
hacla.ml(showopts)
'----- MAKE OUTPUT GAP
hacla.makestates(t=smooth) *shc
hacla.makestates(t=smoothse) *shcse
genr gap_hc = zshc
genr gap_hcp2se=gap_hc+2*zshcse
genr gap_hcm2se=gap_hc-2*zshcse

smpl 1966 1997

```